Flash 5/MX lab: Skiing game (counting hits of rocks and trees)

This lab will provide ideas and methods for implementing a 'thumb-twitching' game in which the player guides a skier down a mountain, moving left and right to avoid obstacles and keeping on-course.  

The player presses the left and right arrow keys to move the skier.  The system keeps track of the number of times the skier hits an obstacle.  The game is over when the skier moves off course to the right or the left or at the base of the mountain.

The critical requirements for you, the program, are to 

· construct the mountain, that is, populate the mountain with obstacles.  The mountain is a tall rectangle (see next point).  The obstacles are placed by creating two movie clips representing a rock and a tree, respectively, and then duplicating each of these a set number of times.  The original rock and tree are often referred to as 'seed' clips.  The duplicates are placed randomly on the stage (on top of the rectangle representing the mountain.

· make the skier move down the mountain.  Your programming will accomplish this by moving the mountain movie clip up the screen.  Only some of the clip will be visible at any time.  It is also necessary to move the obstacles.  The game is over when the mountain has been moved up all the way.

· detect any collisions between skier and obstacle.  This will be done by use of the enterframe clip events for each obstacle.  Note that duplicated movie clips share any clip event code.  The code will call the hitTest method for the corresponding obstacle (rock or tree) with the skier the target argument.  If there is a hit, then the code sets a variable in the obstacle movie clip to true.  If and while this variable is true and the obstacle is hitting the skier, the code will cause the skier to rotate around, an amusing effect.  Use of this variable also checks to make sure that the collision with any one obstacle is not counted more than once.

· respond to the player action of hitting the arrow keys to make the skier move left or right.  The keydown clip event is exactly what is required here.  Your code for handling this event will determine which key was pressed, move the skier over in the indicated direction and check if the skier is off-course.  When the latter happens, the game is over (see next).

· report to the player when the game is over.  You will place two dynamic text fields on the screen.  One is reserved for indicating that the skier went off course.  The other displays the number of hits (that is, collisions with rocks or trees).

The timeline will consist of 3 layers: board, actions and flags.  The board layer contains all the graphics.  The actions layer is where you will put all the ActionScript.  Note: we recommend that you place functions here to carry out the action for the movie clip events.  This makes the coding practically all in one place.  It will be explained more below.  The flags layer will contain a flag for the second frame.  This also is to make your project well-organized.

This project contains 4 symbols, each 1 frame movie clips, in the Library.  Create your own versions and then place them off stage in the first frame of the board layer.  Code in the first frame will place courseA, create duplicates of rock0 and tree0 and place them all 'on' the course.  Lastly, the skier will be placed 'on' the course.  

	Name
	Instance name
	Description 
	Variables

	courseA
	courseA
	Rectangle: 400 by 1200.  Origin: 0,0. You can decorate this or make it a plain light gray color.
	No variables

	rock
	rock0
	Drawing of a rock. See screen shot above.  A seed movie clip for more rock obstacles.
	var notyethit

	tree
	tree0
	Drawing of a tree. See screen shot above.  A seed movie clip for more tree obstacles.
	var notyethit

	skier
	skier
	Drawing of a skier. See screen shot above.  Make the origin in the center of the figure.  
	No variables


In the board layer, first frame, create two Dynamic Text fields.  Place them in the lower part of the stage.  Name one result1 and the other result2.  These will be set at the end of the game.

Click on the actions layer and create 3 keyframes.  Click on the flags layer and create 3 keyframes. Wait to create the 2 additional frames for the board layer until after you have added the clips to the stage and added the event code.  If you add the frames now, you will need to be sure and add the clips and event code to each frame. In the second keyframe, go to Window/Panel/Frame in Flash 5 and Properties in Flash MX and name the frame continue.  Go back to the actions layer, first frame.

In this first frame of the actions layer, you will set up the necessary variables (global variables, that is, they persist between events and they are accessible) and the functions.  The code to set up the variables is: 

var hits;

var oncourse;

var nl;

var unit;

nl = 0;

oncourse = true;

hits = 0;

unit = 10;

stuff = [rock0,tree0];

The hits variable keeps track of the number of collisions.  The oncourse variable keeps track of whether or not the skier goes off-course.  If the skier does go off-course, then the game stops.  This variable will be examined in the 3rd frame to see if the game is to continue (specifically, go to the continue frame) or stop.  The nl variable keeps track of the levels.  This is used in the creation of obstacles by duplicating the seed movie clips. The unit variable holds the distance that the skier will move with each frame.  More correctly, this is the distance that everything else will move each frame.  I make it a variable to prepare for future enhancement in which you may want to alter the speed of the skier under player control.  Lastly, stuff is initialized here as an array with two elements.  More elements will be added.  The elements are put together, so to speak, in this array to allow code to move them.  

The first frame, actions layer, contains the definition of 3 functions: buildcourse, collision and keytest.  The buildcourse function is invoked here in the first frame.  You could just put in the code and not encapsulate it in a function.  I suggest making it a function to prepare for enhancements.  The collision and keytest functions are invoked in clip events for trees and rocks and for the skier (this will be explained in more detail below).  I suggest using function calls here to keep the code in one place.  By the way, a possible enhancement could involve making the response to colliding (hitting) a tree different than hitting a rock.  

The code for building the course is as follows:

function buildcourse() {

_root.courseA._x =0;

_root.courseA._y = 0; 

tree0._x = 50;

tree0._y = 50;

rock0._x = 0;

rock0._y = 300;

rock0.notyethit = true;

tree0.notyethit = true;

var i;

for (i=1;i<6;i++) {

 var namet;

 var namer;

 namet = "tree" + i;

 namer = "rock" + i;

 tree0.duplicateMovieClip(namet,nl);

 stuff.push(_root[namet]);

 _root[namet].notyethit = true;

 nl++;

 rock0.duplicateMovieClip(namer,nl);

 nl++;

 stuff.push(_root[namer]);

  _root[namer].notyethit = true;

 _root[namet]._x = 20 + Math.Random()* courseA._width;

 _root[namet]._y = Math.Random() * courseA._height;

 _root[namer]._x = 20 + Math.Random()* courseA._width;

 _root[namer]._y = Math.Random() * courseA._height;

}

_root.skier._x = .5 * courseA._width;

_root.skier._y = 50;

_root.skier.swapDepths(nl);

}

buildcourse();

This function can be explained by thinking of it as being in three parts.  In the first part, the courseA (the name is intended to be suggestive of defining other courses), is positioned at the top of the stage.  The lower part of the clip is not visible.  The _y value will be decreased to have the effect of moving the course up (and the skier down).  The existing rock instance, named rock0, and the existing tree instance, named tree0, are also placed on the stage.  

The second part of building the course is creating more obstacles.  This is done through a for loop set arbitrarily to iterate 5 times.  At each iteration, a new tree and a new rock are created by use of the duplicateMovieClip method.  The level parameter is incremented using the variable nl so that no new clips are at the same level, which if not prevented, would erase all but the last clip.  I specify nl as a global variable to prepare for any later enhancement involving more creation of clips.  The notyethit variables of the rock and tree instances are set to true.  Each tree and rock is positioned by calls to the Math.Random function.  This returns a fraction (from 0 to 1).  The positioning is done in terms of the _width and _height of courseA.  Note that this could and generally would position trees and rocks off the stage, depending on the _y value.  The value 20 is added to bias the positioning to not be off stage left. You can certainly experiment with this code.

The last part of the function positions the skier and sets the skier depth to make it on top of the course, but behind all but one obstacle.  Again, you can experiment with different effects.  Note that this does not change the collision detection.

There are two more functions that will go in the first frame.  One, called collision, checks if the skier had collided with an object. If so, the function starts the skier rotating.  The other function is invoked if a key is pressed.  The function determines which key and moves the skier (actually, moves the course and everything in it) to give the impression that the skier has moved.  Here is how you program the call of these two functions.  Click on the first frame of the board layer.  Now click on the rock0 instance.  Make the Action window visible: Click on Window/Action.  Then click the arrow in the upper right and invoke Expert mode.  Write the following:

onClipEvent (enterFrame) {


_root.collision(this);

}

Repeat for the tree0 instance.  What this does is make the handling of clip events for the two original (seed) rock and tree and all duplicates be a call to the collision function you are about to program.  The function takes a parameter: the call sets the parameter to be this, specifically the instance for this movie clip.  All obstacles will call collision in turn, passing references to themselves to be used in the code.

Now click on the skier instance and program this code using the Action window:

onClipEvent (keyDown) {

  _root.keytest();

}

Now we can return to the first frame, actions layer and write these two functions. 

You need to understand (and change if and when you wish) what the action is to be for detecting and acting on collisions. For this project, I want the hit counter to be incremented once and only once for each obstacle if the skier hits that obstacle.  It turns out that there is a method that will detect if there is any overlap of the graphics of a tree or a rock and the skier.  However, we now need to review what we want to have happen.  The obstacles and the skier move relative to one another in small, discrete steps.  It is possible, even probable, that the collision will be repeated for several of these steps.  Once I realized that, I decided that I would make the skier flip around during the time that the obstacle and skier were colliding and then settle back to the original rotation once past the obstacle.  The variable: notyethit starts out as true, is set to false once a hit is detected.  If the condition of colliding and notyethit being false is detected, the skier is incrementally rotated.  Then when no more collision is occurring, the variable is set back to true and the skier is reset to zero rotation.  Here is the code:

function collision(thing) {

if (thing.hitTest(_root.skier)) {

      if (thing.notyethit) {



_root.hits += 1;


    thing.notyethit = false;

       }

      else {



_root.skier._rotation +=20;

      }


}

   else {

    if (thing.notyethit==false) {

     thing.notyethit = true;

     _root.skier._rotation = 0;

   }

  }

}

The formal argument thing holds the reference to the particular tree or rock.  It is thing's variable that is examined and modified.  Note that thing is my term.  The name used in the call, this, is a built-in Flash term for the particular instance associated with an event.  (Other languages, such as C++ and Java, also have a this construct.)

The keytest function is called in the event that a key on the keyboard is pressed. I have made it an event associated with the skier instance, but it could be associated with any movie clip on the stage.  As presently written, it makes two tests, using the getCode method of the built-in Key object. I learned the numbers 39 and 37 from reading a Flash text.  I could also have set up a test.  

function keytest() {

if (Key.getCode() == 39) {

    _root.skier._x += 20;

    if (_root.skier._x>_root.courseA._width) {

    result1 = "OFF COURSE";

     _root.oncourse = false;

    }

   }

   else {

    if (Key.getCode() == 37) {

    _root.skier._x -= 20;

    if (_root.skier._x<0) {

     result1 = "OFF COURSE"

    _root.oncourse = false;

    }

   }

  } 

}

The code checks if the right arrow (code 39) key has been pressed.  If so, the skier is moved to the right (the _x property is incremented). A check is then made if the skier is off-course, defined by its _x value being greater than the _width of courseA.  In this case, the phrase "OFF COURSE" is placed in the result1 dynamic text field and the oncourse variable is set to false.  This variable will be checked in the third frame.  The action for the left arrow is similar.  If and when you want to enhance the game with more player options, you will place code to check for other key codes.  WARNING: there is a possibility that 39 and 37 are not the right codes for a given keyboard.  If this seems to be the problem, you can put in a trace statement:
         trace("key pressed was " + Key.getCode());

and change your code accordingly.

Now we move on to moving the skier, that is, moving everything but the skier!  In the actions layer, frame 2 (also labeled with the flag continue), write the following code:

courseA._y = courseA._y - unit;

var i;

for (i=0;i<stuff.length;i++) {

  stuff[i]._y = stuff[i]._y - unit;

}

The first line moves the courseA up (lower values of the _y property). The for loop moves each element contained (referenced) in the stuff array up.  

Lastly, we come to the third frame.  The code here must determine if the game is to continue (that is, control go back to the frame aptly named continue.  There are two circumstances to end the game: the end of the course (the skier has skied the whole mountain) or the skier has gone off course.  

if (_root.oncourse && ((-courseA._y)< courseA._height)) {

gotoAndPlay ("continue");

 }

else

 {

 result2 = "You were hit " + _root.hits + " times.";

 stop();

}

The code checks oncourse still being true and the positioning of courseA.  The latter is done by negating the _y property of courseA (remember it starts off at zero and gets increasingly more negative) and comparing it to the _height of courseA.  

Now add the two keyframes to the board layer.  The material (rock0, tree0, skier and courseA) and the associated event handlers will be replicated in the new frames.

Try it!  Use the trace function and Movie Explorer to check what you have.

That is it for the basic game.  Possible enhancements include:

· improving the quality of drawing of the skier, rocks and trees.  See below concerning the speed and 'the tuck'.  The graphic representing the course (courseA) could be enhanced, though if this image is too intricate, performance may suffer as it is continually re-drawn.

· using a button to start and stop the skier.  A further enhancement would be to re-start the game, with the option of the same course or a new course.  To implement a new run with the same course, you would need to store the original positions of all the trees and rocks.  

· making hitting an obstacle end the game.  Keep track of how much time (possibly number of frames or number of times the course was advanced by change in courseA._y).

· using keys, possibly the up and down, to control the speed (value of unit).  You should then keep track of the time or frame count in some way, with shorter times being better.  An aesthetic enhancement would be to change the graphic representing the skier to be more or less 'into a tuck' with increases or decreases in speed.

· changing the vertical speed (unit) if and when the skier does go left or right.  You can also transform the graphic so the skis are aimed in the appropriate direction.  Change the skier/course relationship in both dimensions (_x and _y) using trigonometry.

· placing clips representing fences on the course so the course is not rectangular. You can use collision tests in terms of the fences.  One approach is to have a small set of distinct fence shapes and have the clip event handlers for all invoke the same function.

· placing clips representing gates on the course.  This is done in the two other ski examples, one of which has a tutorial.  You can start thinking about it before reading the tutorial.  
· making the course of infinite length.  This could be done by continually re-positioning courseA below itself, so to speak.  It may be best done with a courseA and a courseB.  With an unlimited length course, you need a way to end the game, perhaps after a certain number of hits or under player control.

